HEWLETT-PACKARD

HP-41C

USERS' LIBRARY SOLUTIONS

Test Statistics



NOTICE

INTRODUCTION

This HP-41C Solutions book was written to help you get the most from your calculator. The programs were chosen to provide useful calculations for many of the common problems encountered.

They will provide you with immediate capabilities in your everyday calculations and you will find them useful as guides to programming techniques for writing your own customized software. The comments on each program listing describe the approach used to reach the solution and help you follow the programmer's logic as you become and expert on your HP calculator.

KEYING A PROGRAM INTO THE HP-41C

There are several things that you should keep in mind while you are keying in programs from the program listings provided in this book. The output from the HP 82143A printer provides a convenient way of listing and an easily understood method of keying in programs without showing every keystroke. This type of output is what appears in this handbook. Once you understand the procedure for keying programs in from the printed listings, you will find this method simple and fast. Here is the procedure:

1. At the end of each program listing is a listing of status information required to properly execute that program. Included is the SIZE allocation required. Before you begin keying in the program, press [XEQ] [ALPHA] SIZE [ALPHA] and specify the allocation (three digits; e.g., 10 should be specified as 010).

   Also included in the status information is the display format and status of flags important to the program. To ensure proper execution, check to see that the display status of the HP-41C is set as specified and check to see that all applicable flags are set or clear as specified.

2. Set the HP-41C to PRGM mode (press the [PRGM] key) and press ■ [GTO] [•] [•] to prepare the calculator for the new program.

3. Begin keying in the program. Following is a list of hints that will help you when you key in your programs from the program listings in this handbook.

   a. When you see " (quote marks) around a character or group of characters in the program listing, those characters are ALPHA. To key them in, simply press [ALPHA], key in the characters, then press [ALPHA] again. So "SAMPLE" would be keyed in as [ALPHA] "SAMPLE" [ALPHA].

   b. The diamond in front of each LBL instruction is only a visual aid to help you locate labels in the program listings. When you key in a program, ignore the diamond.

   c. The printer indication of divide sign is /. When you see / in the program listing, press [÷].

   d. The printer indication of the multiply sign is ⁑ . When you see ⁑ in the program listing, press [×].

   e. The ⊢ character in the program listing is an indication of the [APPEND] function. When you see ⊢, press ■ [APPEND] in ALPHA mode (press ■ and the K key).

   f. All operations requiring register addresses accept those addresses in these forms:

      nn (a two-digit number)
      IND nn (INDIRECT: ■ , followed fy a two-digit number)
      X, Y, Z, T, or L (a STACK address: [•] followed by X, Y, Z, T, or L)
      IND X, Y, Z, T or L (INDIRECT stack: ■ [•] followed by X, Y, Z, T, or L)

      Indirect addresses are specified by pressing ■ and then the indirect address. Stack addresses are specified by pressing [•] followed by X, Y, Z, T, or L. Indirect stack addresses are specified by pressing ■ [•] and X, Y, Z, T, or L.

Printer Listing	Keystrokes	Display
01♦LBL "SAMPLE"	■ [LBL] [ALPHA] SAMPLE [ALPHA]	*01 LBL^{T}SAMPLE*
02 "THIS IS A "	[ALPHA] THIS IS A [ALPHA]	*02^{T}THIS IS A*
03 "⊢SAMPLE"	[ALPHA] ■ [APPEND] SAMPLE	*03^T ⊢ SAMPLE*
04 AVIEW	■ AVIEW [ALPHA]	*04 AVIEW*
05 6	6	*05 6*
06 ENTER↑	[ENTER↑]	*06 ENTER ↗*
07 -2	2 [CHS]	*07 -2*
08 /	[÷]	*08 /*
09 ABS	[XEQ] [ALPHA] ABS [ALPHA]	*09 ABS*
10 STO IND L	[STO] ■ [•] L	*10 STO IND L*
11 "R3="	[ALPHA] R3= ■ [ARCL] 03	*11^{T}R3=*
12 ARCL 03	■ [AVIEW]	*12 ARCL 03*
13 AVIEW	[ALPHA]	*13 AVIEW*
14 RTN	■ [RTN]	*14 RTN*

TABLE OF CONTENTS

1. ONE SAMPLE TEST STATISTICS FOR THE MEAN 1

Calculates the z statistic for testing the mean if the variance is known. If the variance is unknown, then the t statistic is calculated.

2. TEST STATISTICS FOR THE CORRELATION COEFFICIENT 5

The t statistic can be used to test if the true correlation coefficient is zero. The z statistic, which can be used to test if the correlation coefficient equals a given number (usually non-zero) is also calculated.

3. DIFFERENCES AMONG PROPORTIONS 9

Calculates the chi-square statistic for testing if several independent binomial distributions have equal means.

4. BEHRENS-FISHER STATISTIC 14

Given random samples form two independent normal populations with unequal variances (unknown), this program calculates the Behren-Fisher statistic for testing the means.

5. KRUSKAL-WALLIS STATISTIC 19

The Kruskal-Wallis statistic can be used to test is the independent random samples come from identical continuous population.

6. MEAN-SQUARE SUCCESSIVE DIFFERENCE 24

The mean-square successive difference is used to test if a given sample is random. Suppose the sample size is large and the population is normal, then a z statistic is used instead.

7. THE RUN TEST FOR RANDOMNESS 28

For a given sequence, the z statistic is calculated for testing the randomness of the sequence.

8. INTRACLASS CORRELATION COEFFICIENT 33

Calculates the intraclass correlation coefficient which measures the degree of association among individuals within classes or groups.

9. FISHER'S EXACT TEST FOR A 2 X 2 CONTINGENCY TABLE 39

Fisher's exact probability test is used to analyze a 2 x 2 contingency table when the two independent samples are small in size.

10. BARTLETT'S CHI-SQUARE STATISTIC 46
This chi-square statistic can be used to test the homogeneity of variances. Error corrector for erroneous input data is provided.

11. MANN-WHITNEY STATISTIC 51
Calculates the Mann-Whitney statistic on two independent samples of equal or unequal sizes. Error corrector for erroneous input data is provided.

12. KENDALL'S COEFFICIENT OF CONCORDANCE 56
Calculates Kendall's coefficient of concordance to test agreement between rankings. Error corrector for erroneous input data is provided.

ONE SAMPLE TEST STATISTICS FOR THE MEAN

Suppose $\{x_1, x_2, \ldots, x_n\}$ is a sample from a normal population with a known variance σ^2 and unknown mean μ. A test of the null hypothesis

$$H_0 : \mu = \mu_0$$

is based on the z statistic which has a standard normal distribution.

If the variance σ^2 is unknown then the t statistic, which has the t distribution with n - 1 degrees of freedom, is used instead.

Equations:

$$z = \frac{\sqrt{n}\,(\bar{x} - \mu_0)}{\sigma}$$

$$t = \frac{\sqrt{n}\,(\bar{x} - \mu_0)}{s}$$

where $\bar{x}$ and s are sample mean and sample standard deviation.

Remark: $n > 1$.

Reference: This program is a translation of the HP-65 Stat Pac 2 program.

Example:

Calculate the z and the t statistics for the following set of data if $\mu_0 = 2$ and $\sigma = 1$.

{2.73, 0.45, 2.52, 1.19, 3.51}

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 009	
[XEQ] [ALPHA] ONEST [ALPHA]	ONE SAMPLE T.
2.73 [Σ+] .45 [Σ+] 2.52 [Σ+]	
1.19 [Σ+] 3.51 [Σ+]	5.00
[R/S]	MU NAUGHT ?
2 [R/S]	SIGMA ?
1 [R/S]	Z=0.18
[R/S]	T=0.14
[R/S]	XBAR=2.08
[R/S]	S=1.24

User Instructions

SIZE: 009

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Key in the program.			
2	Initialize the program.		[XEQ] ONEST	ONE SAMPLE T.
3	Input data. Repeat steps 3-4 for			
	$i = 1, 2, \ldots, n$.	x_i	[Σ+]	(i)
4	If you make a mistake inputting x_k, delete			
	it and go to step 3.	x_k as entered	[Σ-]	(k-1)
5	Input μ_0 and σ and calculate z and t.		[R/S]	MU NAUGHT ?
		μ_0	[R/S]	SIGMA ?
		σ	[R/S]	Z=(z)
			[R/S]	T=(t)
			[R/S]	XBAR=($\bar{x}$)
			[R/S]	S=(s)
6	To calculate z and t for a different pair			
	of μ_0 and σ, go to step 5.			
7	To use the program for another set of			
	data, go to step 2.			

Program Listings

Initialize

```
01♦LBL "ONE
ST"
02 FIX 2
03 CLRG
04 ΣREG 00
05 "ONE SAM
PLE T."
06 AVIEW
07 STOP
```

Store μ_0 and σ and make calculations

```
08♦LBL E
09 "MU NAUG
HT ?"
10 PROMPT
11 STO 06
12 "SIGMA ?
"
13 PROMPT
14 STO 07
15 MEAN
16 RCL 06
17 -
18 RCL 05
19 SQRT
20 *
21 STO 08
22 RCL 07
23 /
24 "Z"
25 XEQ 11
26 SDEV
27 RCL 08
28 X<>Y
29 /
30 "T"
31 XEQ 11
32 MEAN
33 "XBAR"
34 XEQ 11
35 SDEV
36 "S"
37 XEQ 11
38 XEQ E
```

Display subroutine

```
39♦LBL 11
40 "⊢="
41 ARCL X
42 AVIEW
43 STOP
44 RTN
45 .END.
```

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	Σx	50	
	Σx^2		
	Σy		
	Σy^2		
	Σxy		
05	n	55	
	μ_0		
	σ		
	$\sqrt{n}(x-\mu_0)$		
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 009 TOT. REG. 24 USER MODE
ENG ____ FIX 2 SCI ____ ON ____ OFF X
DEG ____ RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

TEST STATISTICS FOR THE CORRELATION COEFFICIENT

Under the assumptions of normal correlation analysis, the t statistic , which has the t distribution with n - 2 degrees of freedom, can be used to test the null hypothesis that the true correlation coefficient $\rho = 0$.

To test the null hypothesis $\rho = \rho_0$, where ρ_0 is a given number, the z statistic is used. z has approximately the standard normal distribution.

Equations:

$$t = \frac{r\sqrt{n-2}}{\sqrt{1-r^2}}$$

$$z = \frac{\sqrt{n-3}}{2} \ln\left[\frac{(1+r)(1-\rho_0)}{(1-r)(1+\rho_0)}\right]$$

where r is an estimate (based on a sample of size n) of the correlation coefficient ρ.

Remarks:

1. This program requires that $n > 3$, $|r| < 1$ and $|\rho_0| < 1$; otherwise "DATA ERROR" will result.
2. Usually, the z statistic is used when the sample size is large.

References:

1. Hogg and Craig, Introduction to Mathematical Statistics, Macmillan and Co., 1970.
2. J. Freund, Mathematical Statistics, Prentice-Hall, 1971.
3. This program is a translation of the HP-65 Stat Pac 2 program.

Example:

Given $r = 0.12$, $n = 31$, and $\rho_0 = 0$, find t and z.

Keystrokes:	Display:
[USER]	(set USER mode)
[XEQ] [ALPHA] SIZE [ALPHA] 003	
[XEQ] [ALPHA] CORRTS [ALPHA]	COR. COEF. T.S.
	N ?
31 [R/S]	R ?
.12 [R/S]	T=0.65
[E]	RHO NAUGHT ?
0 [R/S]	Z=0.64

User Instructions

SIZE: 003

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Key in the program and set USER mode.		[USER]	
2	Initialize the program.		[XEQ] CORRTS	COR. COEF. T.S.
3	To calculate t,			N ?
		n	[R/S]	R ?
		r	[R/S]	T=
4	To calculate z,		[E]	RHO NAUGHT ?
		ρ_0	[R/S]	Z=
5	For a new case, go to step 3 or 4.			

Program Listings

Initialize

```
01+LBL "COR
RTS"
02 FIX 2
03 "COR. CO
EF. T.S."
04 AVIEW
05 PSE
```

n / Test n > 3?

```
06 "N ?"
07 PROMPT
08 STO 01
09 3
10 X<>Y
11 X<=Y?
12 GTO 09
```

r / Test |r| < 1?

```
13 "R ?"
14 PROMPT
15 STO 00
16 XEQ 00
```

Calculate t

```
17 RCL 01
18 2
19 -
20 1
21 RCL 00
22 X↑2
23 -
24 /
25 SQRT
26 RCL 00
27 *
28 "T"
29 GTO 11
```

Test $|\rho_0| < 1$

```
30+LBL E
31 "RHO NAU
GHT ?"
32 PROMPT
33 STO 02
34 XEQ 00
```

Calculate z

```
35 RCL 00
36 1
37 +
38 1
39 RCL 00
40 -
41 /
42 1
43 RCL 02
44 -
45 *
46 1
47 RCL 02
48 +
49 /
50 LN
51 RCL 01
52 3
53 -
54 SQRT
55 *
56 2
57 /
58 "Z"
```

Display routine

```
59+LBL 11
60 "⊢="
61 ARCL X
62 AVIEW
63 STOP
64 RTN
```

Test r and ρ_0

```
65+LBL 00
66 ABS
67 1
68 X<>Y
69 X>Y?
70 GTO 09
71 RTN
```

Generate "DATA ERROR"

```
72+LBL 09
73 0
74 /
75 .END.
```

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	r	50	
	n		
	ρ_0		
05		55	
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 003 TOT. REG. 21 USER MODE

ENG ___ FIX 2 SCI ___ ON X OFF ___

DEG ___ RAD ___ GRAD ___

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

DIFFERENCES AMONG PROPORTIONS

Suppose x_1, x_2, ..., x_k are observed values of a set of independent random variables having binomial distributions with parameters n_i and θ_i ($i = 1, 2, \ldots, k$).

A chi-square statistic χ^2 can be used to test the null hypothesis $\theta_1 = \theta_1 = \ldots = \theta_k$. The χ^2 statistic has the chi-square distribution with $k - 1$ degrees of freedom.

Equations:

$$\chi^2 = \sum_{i=1}^{k} \frac{(x_i - n_i\hat{\theta})^2}{n_i\hat{\theta}(1 - \hat{\theta})} = \sum_{i=1}^{k} n_i \left[\frac{1}{\sum_{i=1}^{k} x_i} \sum_{i=1}^{k} \frac{x_i^2}{n_i} + \frac{1}{\sum_{i=1}^{k} (n_i - x_i)} \sum_{i=1}^{k} \frac{(n_i - x_i)^2}{n_i} - 1 \right]$$

where

$$\hat{\theta} = \sum_{i=1}^{k} x_i \Big/ \sum_{i=1}^{k} n_i$$

References: 1. J. Freund, Mathematical Statistics, Prentice-Hall, 1971.
2. This program is a translation of the HP-65 State Pac 2 program.

Example:

	n_i	x_i
Sample 1	400	232
Sample 2	500	260
Sample 3	400	197

Keystrokes:	Display:
[USER]	(set USER mode)
[XEQ] [ALPHA] SIZE [ALPHA] 010	
[XEQ] [ALPHA] DIFF [ALPHA]	DIFF. A. PROPS
	N1 ?
400 [R/S]	X1 ?
232 [R/S]	N2 ?
500 [R/S]	X2 ?
260 [R/S]	N3 ?
400 [R/S]	X3 ?
197 [R/S]	N4 ?
[E]	CHI-SQ=6.47
[R/S]	dF=2.00
[R/S]	THETA=0.53

User Instructions

SIZE: 010

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Key in the program and set USER mode.		[USER]	
2	Initialize the program.		[XEQ] DIFF	DIFF. A. PROPS
3	Input data. Repeat steps 3-4 for			N1 ?
	$i = 1, 2, \ldots, n$.	n_i	[R/S]	X(i)?
		x_i	[R/S]	N(i+1)?
4	If you make a mistake inputtine n_k or x_k,		[C]	N(K)?
	delete the incorrect entry and go back to	n_k as entered	[R/S]	X(K)?
	step 3.	x_k as entered	[R/S]	N(K)?
5	Calculate χ^2.		[E]	CHI-SQ=(χ^2)
6	Calculate df.		[R/S]	dF=(df)
7	Calculate $\hat{\theta}$.		[R/S]	THETA=($\hat{\theta}$)
8	To use the program for another set of data,			
	go to step 2.			

Program Listings

01◆LBL "DIF F" 02 FIX 2 03 CLRG 04 CF 00 05 CF 29 06 "DIFF. A PROPS" 07 AVIEW 08 PSE 09 GTO A	Initialize
10◆LBL C 11 SF 00	For corrections
12◆LBL A 13 1 14 FS? 00 15 CHS 16 ST+ 03 17 "N" 18 XEQ 12 19 STO 07	n_i
20 "X" 21 XEQ 12 22 STO 08	x_i
23 FS? 00 24 CHS 25 ST+ 01 26 ABS 27 - 28 STO 04 29 FS? 00 30 CHS 31 ST+ 02 32 ABS 33 RCL 08 34 + 35 STO 09 36 RCL 08 37 X↑2 38 X<>Y 39 / 40 FS? 00 41 CHS 42 ST+ 05 43 ABS 44 RCL 04 45 X↑2 46 RCL 09 47 / 48 FS? 00 49 CHS 50 ST+ 06 51 FC?C 00 52 GTO A 53 1 54 ST- 03 55 GTO A	accumulate sums
56◆LBL E 57 RCL 05 58 RCL 01 59 / 60 RCL 06 61 RCL 02 62 / 63 + 64 1 65 - 66 RCL 01 67 RCL 02 68 + 69 * 70 "CHI-SQ" 71 XEQ 11	Calculate y^2
72 RCL 03 73 2 74 - 75 "dF" 76 XEQ 11	Calculate df
77 RCL 01 78 RCL 01 79 RCL 02 80 + 81 / 82 "THETA"	Calculate $\hat{\theta}$
83◆LBL 11 84 "⊢=" 85 ARCL X 86 AVIEW 87 STOP 88 RTN	Display result routine
89◆LBL 12 90 FIX 0 91 "⊢" 92 ARCL 03 93 "⊢ ?" 94 AVIEW 95 FIX 2 96 STOP 97 RTN 98 .END.	Display input routine
00	

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00		50	
	Σx_i		
	$\Sigma\ (n_i - x_i)$		
	k		
	$n_i - x_i$		
05	$\Sigma\ (x_i^2/n_i)$	55	
	$\Sigma\ (n_i - x_i)^2/n_i$		
	n_i		
	x_i		
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 010 TOT. REG. 35 USER MODE

ENG ___ FIX 2 SCI ___ ON X OFF ___

DEG ___ RAD ___ GRAD ___

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
00		For correction	Normal
29		For proper display format	

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

BEHRENS-FISHER STATISTIC

Suppose $\{x_1, x_2, \ldots, x_{n_1}\}$ and $\{y_1, y_2, \ldots, y_{n_2}\}$ are independent random samples from two normal populations having means μ_1, μ_2 (unknown). If the variances σ_1^2, σ_2^2 cannot be assumed equal, then the Behrens-Fisher statistic d is used instead of the t statistic to test the null hypothesis

$$H_0 : \mu_1 - \mu_2 = D$$

Equation:

$$d = \frac{\bar{x} - \bar{y} - D}{\sqrt{\frac{s_1^2}{n_1} + \frac{s_2^2}{n_2}}}$$

where $\bar{x}$, $\bar{y}$ and s_1^2, s_2^2 are sample means and variances.

Critical values of this test are tabulated in the Fisher-Yates Tables for various values of n_1, n_2, α and θ, where α is the level of significance and

$$\theta = \tan^{-1}\left(\frac{s_1}{s_2}\sqrt{\frac{n_2}{n_1}}\right)$$

Remark: $n_1 > 1$, $n_2 > 1$.

References:

1. Fisher and Yates, Statistical Tables for Biological, Agricultural and Medical Research, Hafner, Publishing Co., 1970.
2. This program is a translation of the HP-65 Stat Pac 2 program.

Example:

Calculate the Behrens-Fisher statistic for D = 0.

x: 79, 84, 108
y: 91, 103, 90, 113, 108

Keystrokes:	Display:
[USER]	(set USER mode)
[XEQ] [ALPHA] SIZE [ALPHA] 010	
[XEQ] [ALPHA] BEH [ALPHA]	BEHRENS-FISH.
79 [Σ+] 84 [Σ+] 108 [Σ+]	3.00
[R/S]	XBAR=90.33
[R/S]	S2/N=80.11
91 [Σ+] 103 [Σ+] 90 [Σ+] 113 [Σ+]	
108 [Σ+]	5.00
[R/S]	YBAR=101.00
[R/S]	S2/N=20.90
[E]	D ?
0 [R/S]	d=-1.06
[R/S]	THETA=62.94

User Instructions

SIZE: 010

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Key in the program and set USER mode.		[USER]	
2	Initialize the program.		[XEQ] BEH	BEHRENS-FISH
3	Input x data. Repeat steps 3-4 for			
	$i = 1, 2, \ldots, n_1$.	x_i	[Σ+]	(i)
4	If you make a mistake inputting x_k, delete			
	it and go to step 3.	x_k as entered	[Σ-]	(k-1)
5	Calculate $\bar{x}$.		[R/S]	XBAR=($\bar{x}$)
6	Calculate s_1^2/n		[R/S]	S2/N=(s_1^2/n)
7	Input y data. Repeat steps 7-8 for			
	$j = 1, 2, \ldots, n_2$.	y_i	[Σ+]	(j)
8	If you make a mistake inputting y_h, delete			
	it and go to step 7.	y_h as entered	[Σ-]	(h-1)
9	Calculate $\bar{y}$.		[R/S]	YBAR=($\bar{y}$)
10	Calculate s_2^2/n		[R/S]	S2/N=(s_2^2/n)
11	Calculate d.		[E]	D ?
		D	[R/S]	d=(d)
12	Calculate θ.		[R/S]	THETA=(θ)
13	For a different D, go to step 11.			
14	To use the program for other sets of data,			
	go to step 2.			

Program Listings

01♦LBL "BEH" 02 FIX 2 03 CLRG 04 CF 01 05 ΣREG 00 06 "BEHRENS-FISH." 07 AVIEW 08 STOP	Initialize
09♦LBL 05 10 MEAN 11 FS? 01 12 GTO 02 13 STO 06 14♦LBL 02 15 STO 08 16 "XBAR" 17 FS? 01 18 XEQ 01 19 XEQ 11	Calculate and display $\bar{x}$, $\bar{y}$
20 SDEV 21 X↑2 22 RCL 05 23 / 24 FS? 01 25 GTO 03 26 STO 07 27♦LBL 03 28 STO 09 29 CLΣ 30 SF 01 31 "S2/N" 32 XEQ 11 33 GTO 05	Calculate and display s_i^2/N
34♦LBL 01 35 "YBAR" 36 RTN	
37♦LBL E 38 "D ?" 39 PROMPT 40 CHS 41 RCL 08 42 - 43 RCL 06 44 + 45 RCL 07 46 RCL 09 47 + 48 SQRT 49 / 50 "d" 51 XEQ 11	Calculate d
52 RCL 07 53 RCL 09 54 / 55 SQRT 56 ATAN 57 "THETA"	Calculate θ
58♦LBL 11 59 "⊢=" 60 ARCL X 61 CLX 62 AVIEW 63 STOP 64 RTN 65 .END.	Display result routine

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	used in summations	50	
	used in summations		
	used in summations		
	used in summations		
	used in summations		
05	used in summations	55	
	$\bar{x}$		
	s_1^2/n		
	$\bar{y}$		
	s_2^2/n		
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 010 TOT. REG. 28 USER MODE
ENG ____ FIX 2 SCI ____ ON X OFF ____
DEG X RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
01		"YBAR"	"XBAR"

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

KRUSKAL-WALLIS STATISTIC

Suppose we want to test the null hypothesis that k independent random samples of sizes n_1, n_2, ..., n_k come from identical continuous populations.

Arrange all values from k samples jointly (as if they were one sample) in an increasing order of magnitude. Let R_{ij} ($i = 1, 2, \ldots, k$, $j = 1, 2, \ldots, n_i$) be the rank of the jth value in the ith sample.

The Kruskal-Wallis statistic H can be used to test the null hypothesis.

When all sample sizes are large (>5), H is distributed approximately as the chi-square with k - 1 degrees of freedom. For small samples, the test is based on special tables.

Equation:

$$H = \frac{12}{N(N+1)} \sum_{i=1}^{k} \frac{\left(\sum_{j=1}^{n_i} R_{ij}\right)^2}{n_i} - 3(N+1)$$

where

$$N = \sum_{i=1}^{k} n_i$$

References:
1. W.J. Conover, Practical Nonparametric Statistics, John Wiley and Sons, 1971.
2. Table for small samples (k = 3): Alexander and Quade, On the Kruskal-Wallis Three Sample H-statistic, University of North Carolina, Department of Biostatistics, Inst. Statistics Mimeo Ser. 602, 1968.
3. This program is a translation of the HP-65 Stat Pac 2 program.

Example:

Ranks R_{ij}

i \ j	1	2	3	4	5	6	7	8	9	10
1	29	5	26	10	33	30				
2	11	12	9	7	20	18	19	21		
3	14	28	8	25	17	15	32	4	2	
4	6	27	3	16	24	13	1	31	22	23

Keystrokes:	Display:
[USER]	(set USER mode)
[XEQ] [ALPHA] SIZE [ALPHA] 006	
[XEQ] [ALPHA] KRU [ALPHA]	KRUSKAL-WALL.
	R1,1 ?
29 [R/S]	R1,2 ?
5 [R/S]	R1,3 ?
26 [R/S]	R1,4 ?
⋮	⋮
30 [R/S]	R1,7 ?
[B]	R2,1 ?
11 [R/S]	R2,2 ?
12 [R/S]	R2,3 ?
⋮	⋮
21 [R/S]	R2,9 ?
[B]	R3,1 ?
14 [R/S]	R3,2 ?
28 [R/S]	R3,3 ?
⋮	⋮
2 [R/S]	R3,10 ?
[B]	R4,1 ?
6 [R/S]	R4,2 ?
27 [R/S]	R4,3 ?
⋮	⋮
23 [R/S]	R4,11 ?
[B]	R5,1 ?
[E]	H=2.29
[R/S]	dF=3.00
[R/S]	N=33.00

User Instructions

SIZE: 006

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Key in the program and set USER mode.		[USER]	
2	Initialize the program.		[XEQ] KRU	KRUSKAL-WALL.
3	Perform steps 3-5 for i = 1, 2, ..., k			R1,1 ?
	and j = 1, 2, ..., n_i. Input R_{ij}.	R_{ij}	[R/S]	R(i),(j+1)?
4	If you make a mistake inputting R_{ih},		[C]	R(i),(h)?
	delete it and go to step 3.	R_{ih} as entered	[R/S]	R(i),(h)?
5	For the end of the i'th sample, press		[B]	R(i+1),1?
6	Calculate H,		[E]	H=
	df,		[R/S]	dF=
	and N		[R/S]	N=
7	To use the program for another set of			
	data, go to step 2.			

Program Listings

Initialize

```
01♦LBL "KRU
"
02 CF 29
03 FIX 0
04 CLRG
05 "KRUSKAL
-WALL."
06 AVIEW
07 GTO A
```

Correction

```
08♦LBL C
09 1
10 ST- 01
11 SF 00
```

Input R_{ij}

```
12♦LBL A
13 RCL 01
14 1
15 +
16 RCL 04
17 1
18 +
19 "R"
20 ARCL X
21 "⊢,"
22 ARCL Y
23 "⊢ ?"
24 PROMPT
25 FS? 00
26 CHS
27 ST+ 02
28 1
29 FC?C 00
30 ST+ 01
31 GTO A
```

Compute row i partial results

```
32♦LBL B
33 RCL 01
34 ST+ 05
35 RCL 02
36 X↑2
37 X<>Y
38 /
39 ST+ 03
40 1
41 ST+ 04
42 0
43 STO 01
44 STO 02
45 GTO A
```

Calculate H

```
46♦LBL E
47 FIX 2
48 RCL 03
49 4
50 *
51 RCL 05
52 /
53 RCL 05
54 1
55 +
56 /
57 LASTX
58 -
59 3
60 *
61 "H"
62 XEQ 11
```

Calculate df and N

```
63 RCL 04
64 1
65 -
66 "dF"
67 XEQ 11
68 RCL 05
69 "N"
```

Display routine

```
70♦LBL 11
71 "⊢="
72 ARCL X
73 AVIEW
74 STOP
75 RTN
76 .END.
```

80	
90	
00	

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00		50	
	n_i		
	ΣR_{ij}		
	$\Sigma\ [(\Sigma R_{ij})^2/n_i]$		
	k		
05	N	55	
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 006 TOT. REG. 25 USER MODE

ENG ___ FIX 2 SCI ___ ON X OFF ___

DEG ___ RAD ___ GRAD ___

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
29		For proper display format	

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

MEAN SQUARE SUCCESSIVE DIFFERENCE

When test and estimation techniques are used, the method of drawing the sample from the population is specified to be random in most cases. If observations are chosen in sequence $x_1, x_2, \ldots, x_n$, the mean-square successive difference η can be used to test for randomness.

If the sample size n is large (say, greater than 20) and the population is normal, then a z statistic has approximately the standard normal distribution. Long trends are associated with large positive values of z and short oscillations with large negative values.

Equations:

$$\eta = \sum_{i=2}^{n} (x_i - x_{i-1})^2 \bigg/ \sum_{i=1}^{n} (x_i - \bar{x})^2 = \sum_{i=2}^{n} (x_i - x_{i-1})^2 \Bigg/ \left[\sum_{i=1}^{n} x_i^2 - \frac{\left(\sum_{i=1}^{n} x_i\right)^2}{n} \right]$$

$$z = \frac{1 - \eta/2}{\sqrt{\dfrac{n-2}{n^2-1}}}$$

References: 1. Dixon and Massey, Introduction to Statistical Analysis, McGraw-Hill, 1969.
2. This program is a translation of the HP-65 Stat Pac 2 program.

Example:

Find the mean-square successive difference for the following set of data:

{0.53, 0.52, 0.39, 0.49, 0.97

Keystrokes:	Display:
[USER]	(set USER mode)
[XEQ] [ALPHA] SIZE [ALPHA] 009	
[XEQ] [ALPHA] MNSQD [ALPHA]	MEAN SQ DIFF
.53 [A] .52 [A] .39 [A] .49 [A] .97 [A]	5.00
[E]	ETA=1.27
[R/S]	Z=1.03

User Instructions

SIZE: 009

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Key in the program and set USER mode.		[USER]	
2	Initialize the program.		[XEQ] MNSQD	MEAN SQ DIFF
3	Repeat steps 3-4 for i = 1, 2, ..., n.	x_i	[A]	(i)
4	If you make a mistake inputting x_k, delete			
	it and go to step 3.	x_k	[C]	(k-1)
5	Calculate η		[E]	ETA=(η)
	and z		[R/S]	Z=(z)
6	To use the program for another set of			
	data, go to step 2.			

Program Listings

01♦LBL "MNSQD" 02 FIX 2 03 CLRG 04 SF 01 05 ΣREG 00 06 "MEAN SQ DIFF" 07 AVIEW 08 STOP	Initialize
09♦LBL C 10 RCL 08 11 STO 07 12 RCL 06 13 - 14 RCL 06 15 Σ- 16 STOP	Correction routine
17♦LBL A 18 STO 06 19 RCL 07 20 STO 08 21 - 22 FS?C 01 23 0 24 RCL 06 25 STO 07 26 Σ+ 27 STOP	Compute summations $R_y = x_i - x_{i-1}$
28♦LBL E 29 RCL 03 30 RCL 01 31 RCL 00 32 X↑2 33 RCL 05 34 / 35 - 36 / 37 "ETA" 38 XEQ 11	Calculate η
39 2 40 / 41 1 42 - 43 CHS 44 RCL 05 45 2 46 - 47 RCL 05 48 X↑2 49 1 50 - 51 / 52 SQRT 53 / 54 "Z"	Calculate z
55♦LBL 11 56 "⊦=" 57 ARCL X 58 AVIEW 59 STOP 60 RTN 61 .END.	Display routine

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	Σx_i	50	
	Σx_i^2		
	$\Sigma(x_i - x_{i-1})^2$		
05	n	55	
	x_i		
	x_{i-1}		
	used for corrections		
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 009 TOT. REG. 23 USER MODE
ENG ____ FIX 2 SCI ____ ON X OFF ____
DEG ____ RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
01		for x_1	

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

THE RUN TEST FOR RANDOMNESS

Consider a sequence of symbols such that the symbols are of two types only. A run is a continuous string of identical symbols preceded and followed by a different symbol or no symbol at all. For example, the sequence 1110100011 has five runs.

Let the total number of runs in a given sequence be u, and let n_1 and n_2 represent the number of symbols of type 1 and type 2 respectively. If the sample sizes are large (say, n_1 and n_2 are both greater than 10), then the randomness of the sequence may be tested using a z statistic which has the standard normal distribution.

Equations:

The sample distribution of the run has the mean μ and the standard deviation σ.

$$\mu = \frac{2\, n_1\, n_2}{n_1 + n_2} + 1$$

$$\sigma = \sqrt{\frac{2\, n_1 n_2 (2\, n_1 n_2 - n_1 - n_2)}{(n_1 + n_2)^2 (n_1 + n_2 - 1)}}$$

The test is based on the statistic

$$z = \frac{u - \mu}{\sigma}$$

Remarks:

1. For small samples, the test is based on special tables.
2. This program can also be used for other tests involving runs. For example, one might want to test runs of scores above and below the median based on the order in which the scores were obtained. In this case, a sequence could be constructed in which each score would be replaced by a 1 if it was above the median or a 0, if below the median. The run test for randomness can then be applied to the sequence of 0's and 1's.

   Another use might be for Wald-Wolfowitz run test, which tests the null hypothesis that two random samples have been drawn from identical populations. The data from both groups are combined into one sequence according to magnitude. Each value may be assigned a 0 or 1 depending on which population it came from, and the run test for randomness then performed on the resulting sequence.

References: 1. Freund and Williams, Dictionary/Outline of Basic Statistics, McGraw-Hill, 1966.

2. This program is a translation of the HP-65 Stat Pac 2 program.

Example:

A statistician sits by the roulette table one night in a Las Vegas casino, suspiciously watching the house rake in stake upon stake. To test the null hypothesis that the sequence of numbers is random, the statistician observes the following sequence of red (R) and black (B) numbers (ignoring 0 and 00):

RRRR B RRR BBBBB RR BBB RR BB RRR

In the sequence are 14 R's, 11 B's and a total of 9 runs. Find the mean and standard deviation of the sampling distribution and the z statistic.

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 009	
[XEQ] [ALPHA] RUNTEST [ALPHA]	RUN TEST
	NO. OF RUNS?
9 [R/S]	NO. OF TYPE1?
14 [R/S]	NO. OF TYPE2?
11 [R/S]	MU=13.32
[R/S]	SIGMA=2.41
[R/S]	Z=-1.79

(His suspicion is not entirely unjustified).

User Instructions

SIZE: 009

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Key in the program.			
2	Initialize the program.		[XEQ] RUNTEST	RUN TEST
				NO. OF RUNS ?
3	Key in the number of runs.	u	[R/S]	NO. OF TYPE1 ?
4	Key in the number of type 1.	n_1	[R/S]	NO. OF TYPE2 ?
5	Key in the number of type 2.	n_2	[R/S]	MU=(μ)
			[R/S]	SIGMA=(σ)
			[R/S]	Z=(z)
6	For another case, go to step 2.			

Program Listings

Program	Comments
01♦LBL "RUN TEST" 02 FIX 2 03 "RUN TEST" 04 AVIEW 05 PSE	Initialize
06 "NO. OF RUNS ?" 07 PROMPT 08 STO 03	u
09 "NO. OF TYPE1?" 10 PROMPT 11 STO 01	n_1
12 "NO. OF TYPE2?" 13 PROMPT 14 STO 02	n_2
15 * 16 2 17 * 18 STO 07 19 RCL 01 20 RCL 02 21 + 22 STO 08 23 / 24 1 25 + 26 STO 04 27 "MU" 28 XEQ 11	Calculate μ
29 RCL 07 30 RCL 08 31 - 32 RCL 07 33 * 34 RCL 08 35 ENTER↑ 36 * 37 RCL 08 38 1 39 - 40 * 41 / 42 SQRT 43 STO 05 44 "SIGMA" 45 XEQ 11	Calculate σ
46 RCL 03 47 RCL 04 48 - 49 RCL 05 50 / 51 STO 06 52 "Z"	Calculate z
53♦LBL 11 54 "⊢=" 55 ARCL X 56 AVIEW 57 STOP 58 RTN 59 .END.	Display routine

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00		50	
	n_1		
	n_2		
	u		
	μ		
05	σ	55	
	z		
	$2n_1n_2$		
	n_1+n_2		
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 009 TOT. REG. 28 USER MODE

ENG ____ FIX 2 SCI ____ ON ____ OFF X

DEG ____ RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

INTRACLASS CORRELATION COEFFICIENT

The intraclass correlation coefficient r_I measures the degree of association among individuals within classes or groups.

		Observations			
Groups	1	x_{11}	x_{12}	...	x_{1n}
	2	x_{21}	x_{22}	...	x_{2n}
	.	.	.		.
	.	.	.		.
	.	.	.		.
	k	x_{k1}	x_{k2}	...	x_{kn}

The coefficient is most easily calculated using the analysis of variance techniques. r_I is the sample estimate of the population intraclass correlation coefficient ρ_I. If we can assume that the individuals within groups are random samples from normal populations with the same variance, then the hypothesis $\rho_I = 0$ can be tested using the F statistic.

Equations:

1. Sums

   Group
$$T_i = \sum_{j=1}^{n} x_{ij} \quad i = 1, 2, \ldots, k$$

   Total
$$T = \sum_{i=1}^{k} T_i$$

2. Sums of squares

   Mean
$$MSS = T^2/k\,n$$

   Among groups
$$ASS = \sum_{i=1}^{k} T_i{}^2/n - MSS$$

   Within groups
$$WSS = \sum_{i=1}^{k}\sum_{j=1}^{n} x_{ij}{}^2 - MSS - ASS$$

3. Intraclass correlation coefficient

$$r_I = \left(\frac{ASS}{k-1} - \frac{WSS}{k(n-1)}\right) \div \left(\frac{ASS}{k-1} + \frac{WSS}{k}\right)$$

4. F statistic

$$F = \frac{ASS}{k-1} \div \frac{WSS}{k(n-1)}$$

with $df_1 = k - 1$ and $df_2 = k(n - 1)$ degrees of freedom.

References:
1. B. Ostle, Statistics, in Research, Iowa State University Press, 1972.
2. This program is a translation of the HP-65 Stat Pac 2 program.

Example:

Groups		Observations	
	1	71	71
	2	69	72
	3	59	65
	4	65	64
	5	66	60
	6	73	72
	7	68	67
	8	70	68

Keystrokes:	Display:
[USER]	(set USER mode)
[XEQ] [ALPHA] SIZE [ALPHA] 010	
[XRQ] [ALPHA] INT [ALPHA]	INTRACLASS C.
	N ?
2 [R/S]	X1,1 ?
71 [R/S]	X1,2 ?
71 [R/S]	T1=142
[R/S]	X2,1 ?
69 [R/S]	X2,2 ?
72 [R/S]	T2=141
⋮	⋮
70 [R/S]	X8,2 ?
68 [R/S]	T8=138
[E]	RI=0.70
[R/S]	F=5.61
[R/S]	dF1=7.00
[R/S]	dF2=8.00

User Instructions

SIZE: 010

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Key in the program and set USER mode.		[USER]	
2	Initialize the program.		[XEQ] INT	INTRACLASS C.
				N ?
3	Input n (the number of columns).	n	[R/S]	X1,1 ?
4	Perform steps 4-5 for i = 1, 2, ..., k	x_{ij}	[R/S]	X(i),(j+1)?
	and j = 1, 2, ..., n. T_i is automatically			Ti=(Ti)
	displayed when x_{in} is input. Press		[R/S]	X(i+1),I ?
	[R/S] to continue.			
5	Is you make a mistake inputting x_{ih},		[C]	X(i),(h)?
	correct it and go to step 4 (x_{in} cannot be	x_{ih} as entered	[R/S]	X(i),(h)?
	corrected -- go to step 2).			
6	Calculate r_I,		[E]	RI=(r_I)
	F,		[R/S]	F=(F)
	and the degrees of freedom.		[R/S]	dF1=(df_1)
			[R/S]	dF2=(df_2)
7	For another set of data, go to step 2.			

Program Listings

```
 01♦LBL "INT
"
 02 FIX 0
 03 CLRG
 04 CF 29
 05 CF 00                 Initialize
 06 "INTRACL
ASS C."
 07 AVIEW
 08 PSE
 09 "N ?"
 10 PROMPT
 11 STO 09
 12 GTO a
------------------------------------------
 13♦LBL C
 14 SF 00                 Correction
 15 1                     routine
 16 ST- 01
------------------------------------------
 17♦LBL a
 18 RCL 01
 19 1
 20 +
 21 RCL 02                Input prompt
 22 1                     routine
 23 +
 24 "X"
 25 ARCL X
 26 "⊢,"
 27 ARCL Y
 28 "⊢ ?"
 29 PROMPT
------------------------------------------
 30 FS? 00
 31 CHS
 32 ST+ 06
 33 X↑2
 34 FS? 00
 35 CHS
 36 ST+ 05
 37 1
 38 FC?C 00
 39 ST+ 01
 40 RCL 09                n
 41 RCL 01                j
 42 X≠Y?
 43 GTO a
 44 RCL 06
 45 STO 08                Calculate T_i
 46 ST+ 03
 47 X↑2
 48 ST+ 04
 49 0
 50 STO 01
 51 STO 06
 52 1
 53 ST+ 02
 54 RCL 08
 55 "T"
 56 ARCL 02
 57 XEQ 11
 58 GTO a
------------------------------------------
 59♦LBL E
 60 FIX 2
 61 RCL 04
 62 RCL 03
 63 X↑2
 64 RCL 02
 65 /
 66 -
 67 RCL 09
 68 STO 01
 69 /                     ASS
 70 RCL 02
 71 1                     Calculate r_I
 72 -
 73 /
 74 STO 00
 75 RCL 05
 76 RCL 04
 77 RCL 01
 78 /
 79 -
 80 RCL 02
 81 /
 82 STO 08                WSS/k
 83 RCL 01
 84 1
 85 -
 86 STO 01
 87 /
 88 -
 89 RCL 00
 90 RCL 08
 91 +
 92 /
 93 "RI"
 94 XEQ 11
------------------------------------------
 95 RCL 00
 96 RCL 08
 97 RCL 01                Calculate F
 98 /
 99 /
100 "F"
------------------------------------------
```

Program Listings

```
101 XEQ 11
102 RCL 02
103 1
104 -
105 "dF1"
106 XEQ 11
107 RCL 01
108 RCL 02
109 *
110 "dF2"
111•LBL 11
112 "⊢="
113 ARCL X
114 AVIEW
115 STOP
116 RTN
117 .END.
```

Calculate df_i

Display routine

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	ASS/k-1	50	
	j → n		
	i → k		
	T		
	ΣT_i^2		
05	$\Sigma\Sigma x_{ij}^2$	55	
	Σx_{ij}		
	temp, WSS/k		
	n, n-1		
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 010 TOT. REG. 37 USER MODE

ENG ____ FIX 2 SCI ____ ON X OFF ____

DEG ____ RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
00		For corrections	Normal
29		For proper display format	

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

FISHER'S EXACT TEST FOR A 2 x 2 CONTINGENCY TABLE

Fisher's exact probability test is used for analyzing a 2 x 2 contingency table when the two independent samples are small in size.

a	b
c	d

Suppose a, b, c, d are the frequencies and a is the smallest frequency, this program calculates the following:

1. The exact probability p_0 of observing the given frequencies in a 2 x 2 table, when the marginal totals are regarded as fixed.

2. The exact probability p_i (i = 1, 2, ..., a) of each more extreme table having the same marginal totals.

3. The sum S_i of the probabilities of the first i + 1 tables.

4. The sum S of the probabilities of all tables with the same margins (i.e., $S = S_a$).

Equations:

1.

$$p_0 = \frac{(a+b)!(c+d)!(a+c)!(b+d)!}{N!a!b!c!d!}$$

where

$$N = a + b + c + d.$$

2. For the more extreme table (with the same margins)

a - i	b + i
c + i	d - i

$$p_i = \frac{(a+b)!(c+d)!(a+c)!(b+d)!}{N!(a-i)!(b+i)!(c+i)!(d-i)!}$$

where

i can be 1, 2, ... or a.

3.

$$S_n = \sum_{i=0}^{n} p_i$$

where

n can be 1, 2, ..., a.

4.

$$S = \sum_{i=0}^{a} p_i$$

Remarks: 1. a must be the smallest among the frequencies. Rearrange the table if necessary.

2. This program requires $N \leqslant 69$. However, Fisher's exact test is normally used for $N \leqslant 30$.

References: 1. S. Siegel, Nonparametric Statistics, McGraw-Hill, 1956.

2. Sir R. A. Fisher, Statistical Methods for Research Workers, Oliver and Boyd, 1950.

3. This program is a translation of the HP-65 Stat Pac 2 program.

Example:

Calculate p_0, p_1, p_2, S_4 and S for the following table

7	10
8	5

Note:

The table must be rearranged as

5	8
10	7

Keystrokes:	Display:
[USER]	(set USER mode)
[XEQ] [ALPHA] SIZE [ALPHA] 009	
[XEQ] [ALPHA] FIS [ALPHA]	FISHERS TEST
	a?
5 [R/S]	b?
8 [R/S]	c?
10 [R/S]	d?
7 [R/S]	P0=0.16
[A]	P1=0.06
[A]	P2=0.01
[A] [A] [R/S]	S4=0.23
[E]	S=0.23

User Instructions

SIZE: 009

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Key in the program and set USER mode.		[USER]	
2	Initialize the program.		[XEQ] FIS	FISHERS TEST
				a?
3	Input frequencies and calculate P.	a	[R/S]	b?
		b	[R/S]	c?
		c	[R/S]	d?
		d	[R/S]	PO=(P_o)
4	(Optional) Perform steps 4-5 for			
	i = 1, 2, ..., a. Calculate P_i.		[A]	Pi=(P_i)
5	Calculate S_i.		[R/S]	Si=(S_i)
6	Calculate the sum of all probabilities.		[E]	S=(S)
7	For another set of data, go to step 2.			

Program Listings

Initialize

```
01♦LBL "FIS
"
02 FIX 2
03 CF 01
04 CF 29
05 "FISHERS
TEST"
06 AVIEW
07 PSE
08 CLRG
```

Store a, b, c, d and calculate numerator of P_i

```
09 "a?"
10 PROMPT
11 STO 01
12 STO 08
13 "b?"
14 PROMPT
15 STO 02
16 +
17 STO 05
18 "c?"
19 PROMPT
20 STO 03
21 "d?"
22 PROMPT
23 STO 04
24 +
25 STO 06
26 FACT
27 RCL 05
28 FACT
29 *
30 RCL 05
31 RCL 06
32 +
33 FACT
34 /
35 RCL 01
36 RCL 03
37 +
38 FACT
39 *
40 RCL 02
41 RCL 04
42 +
43 FACT
44 *
45 STO 07
46 0
47 STO 05
48 RDN
```

Loop to calculate P_i

```
49♦LBL 00
50 RCL 01
51 FACT
52 /
53 RCL 02
54 FACT
55 /
56 RCL 03
57 FACT
58 /
59 RCL 04
60 FACT
61 /
62 ST+ 05
63 FS? 01
64 RTN
65 "P"
66 XEQ 11
```

Display S_i

```
67 RCL 05
68 "S"
69 XEQ 11
70 STOP
```

Set up to calculate P_{i+1}

```
71♦LBL A
72 1
73 ST- 01
74 ST+ 02
75 ST+ 03
76 ST- 04
77 ST- 08
78 ST+ 00
79 RCL 07
80 GTO 00
```

Calculate S

```
81♦LBL E
82 SF 01
83 RCL 08
84 0
85 X=Y?
86 XEQ 01
87 XEQ A
88 GTO E
```

Display S

```
89♦LBL 01
90 CF 01
91 RCL 05
92 "S="
93 ARCL X
94 AVIEW
95 STOP
```

Display routine

```
96♦LBL 11
97 FIX 0
98 ARCL 00
99 "├="
100 FIX 2
```

Program Listings

```
101 ARCL X
102 AVIEW
103 STOP
104 RTN
105 .END.
```

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	i	50	
	a		
	b		
	c		
	d		
05	a+b, S_n	55	
	c+d		
	numerator of P_i		
	a		
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 009 TOT. REG. 33 USER MODE
ENG ___ FIX 2 SCI ___ ON X OFF ___
DEG ___ RAD ___ GRAD ___

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
01		Calculate S	Normal
29		Cleared for proper display format	

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

BARTLETT'S CHI-SQUARE STATISTIC

$$\chi^2 = \frac{f \ln s^2 - \sum_{i=1}^{k} f_i \ln s_i^2}{1 + \frac{1}{3(k-1)}\left[\left(\sum_{i=1}^{k} \frac{1}{f_i}\right) - \frac{1}{f}\right]}$$

where: s_i^2 = sample variance of the ith sample

f_i = degrees of freedom associated s_i^2

i = 1, 2, ..., k

k = number of samples

$$s^2 = \frac{\sum_{i=1}^{k} f_i s_i^2}{f}$$

$$f = \sum_{i=1}^{k} f_i$$

This χ^2 has a chi-square distribution (approximately) with k - 1 degrees of freedom which can be used to test the null hypothesis that $s_1^2, s_2^2, \ldots, s_k^2$ are all estimates of the same population variance σ^2; i.e., H_0: Each of $s_1^2, s_2^2, \ldots, s_k^2$ is an estimate of σ^2.

References: 1. Statistical Theory with Engineering Applications, A. Hald, John Wiley and Sons, 1960.

2. This program is a translation of the HP-65 Stat Pac 1 program.

Example:

Apply the program to the following data:

i	1	2	3	4	5	6
s_i^2	5.5	5.1	5.2	4.7	4.8	4.3
f_i	10	20	17	18	8	15

Keystrokes:	Display:
[USER]	(set USER mode)
[XEQ] [ALPHA] SIZE [ALPHA] 009	
[XEQ] [ALPHA] BAR [ALPHA]	BARTLETTS
	F1?
10 [R/S]	S1 SQ?
5.5 [R/S]	F2?
⋮	⋮
15 [R/S]	S6SQ?
4.3 [R/S]	F7?
[E]	CHI SQ=0.25
[R/S]	dF=5.00

User Instructions

SIZE: 009

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Key in the program and set USER mode.		[USER]	
2	Initialize the program.		[XEQ] BAR	BARTLETTS
3	Perform steps 3-4 for i = 1, 2, ..., k.			F1?
	Input f_i.	f_i	[R/S]	S(i) SQ?
	Input S_i^2.	S_i^2	[R/S]	F(i+1)?
4	If you make a mistake inputting f_h or			
	S_h^2, perform this step and go back to step 3.	f_h or S_h^2 as entered	[C]	F(h)? or S(h) SQ?
5	Calculate χ^2		[E]	CHI SQ=(χ^2)
	and df.		[R/S]	dF=(df)
6	To use the program for another set of			
	data, go to step 2.			

Program Listings

Initialize

```
01◆LBL "BAR
"
02 FIX 0
03 CLRG
04 CF 01
05 CF 29
06 "BARTLET
TS"
07 AVIEW
08 PSE
09 GTO A
```

Correct s_i^2

```
10◆LBL C
11 FS? 01
12 GTO 01
13 STO 08
14 RCL 01
15 *
16 ST- 00
17 RCL 08
18 LN
19 RCL 01
20 *
21 ST- 06
22 1
23 ST- 05
24 GTO b
```

Correct f_i

```
25◆LBL 01
26 ST- 03
27 1/X
28 ST- 04
29◆LBL A
30 "F"
31 RCL 05
32 1
33 +
34 ARCL X
35 "⊢?"
36 PROMPT
```

Accumulate sums

```
37 SF 01
38 STO 01
39 ST+ 03
40 1/X
41 ST+ 04
42◆LBL b
43 "S"
44 RCL 05
45 1
46 +
47 ARCL X
48 "⊢ SQ?"
49 PROMPT
50 CF 01
51 STO 08
52 RCL 01
53 *
54 ST+ 00
55 RCL 08
56 LN
57 RCL 01
58 *
59 ST+ 06
60 1
61 ST+ 05
62 GTO A
```

Calculate χ^2 and df

```
63◆LBL E
64 FIX 2
65 RCL 00
66 RCL 03
67 /
68 LN
69 RCL 03
70 *
71 RCL 06
72 -
73 RCL 04
74 RCL 03
75 1/X
76 -
77 RCL 05
78 1
79 -
80 STO 02
81 3
82 *
83 /
84 1
85 +
86 /
87 "CHI SQ"
88 XEQ 11
89 RCL 02
90 "dF"
```

Display routine

```
91◆LBL 11
92 "⊢="
93 ARCL X
94 AVIEW
95 STOP
96 RTN
97 .END.
```

00	

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	$\Sigma f_i s_i^2$	50	
	f_i		
	df		
	Σf_i		
	$\Sigma(1/f_i)$		
05	k	55	
	$\Sigma f_i \ln s_i^2$		
	s_i^2		
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 009 TOT. REG. 32 USER MODE

ENG ___ FIX 2 SCI ___ ON X OFF ___

DEG ___ RAD ___ GRAD ___

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
01		Correct f_i	Correct s_i^2
29		For proper display format	

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

MANN-WHITNEY STATISTIC

This program calculates the Mann-Whitney test statistic on two independent samples of equal or unequal sizes. This test is designed for testing the null hypothesis of no difference between two populations.

Mann-Whitney test statistic is defined as

$$U = n_1 \ n_2 + \frac{n_1(n_1 + 1)}{2} - \sum_{i=1}^{n_1} R_i$$

where n_1 and n_2 are the sizes of the two samples. Arrange all values from both samples jointly (as if they were one sample) in an increasing order of magnitude, let R_i ($i = 1, 2, \ldots, n$) be the ranks assigned to the values of the first sample (it is immaterial which sample is referred to as the "first").

When n_1 and n_2 are small, the Mann-Whitney test bases on the exact distribution of U and specially constructed tables. When n_1 and n_2 are both large (say, greater than 8) then

$$Z = \frac{U - \frac{n_1 n_2}{2}}{\sqrt{n_1 \ n_2(n_1 + n_2 + 1)/12}}$$

is approximately a random variable having the standard normal distribution.

For small samples (say, less than or equal to 8) the specially constructed tables should be used. For example: Handbook of Statistical Tables, D. B. Owen, Addison-Wesley, 1962.

References: 1. Mathematical Statistics, J. E. Freund, Prentice-Hall, 1962.

2. This program is a translation of the HP-65 Stat Pac 1 program.

Example:

Find U and Z for the following data:

Sample 1	14.9	11.3	13.2	16.6	17	14.1	15.4	13	16.9	
Rank R_i	7	1	4	12	14	5	10	3	13	
Sample 2	15.2	19.8	14.7	18.3	16.2	21.1	18.9	12.2	15.3	19.4
Rank	8	18	6	15	11	19	16	2	9	17

NOTE: 1. $n_1 = 9$, $n_2 = 10$

2. The ranks have already been assigned in the example.

Keystrokes:	Display:
[USER]	(set USER mode)
[XEQ] [ALPHA] SIZE [ALPHA] 004	
[XEQ] [ALPHA] MANN [ALPHA]	MANN-WHITNEY
	N1 ?
9 [R/S]	N2 ?
10 [R/S]	R1 ?
7 [R/S]	R2 ?
1 [R/S]	R3 ?
⋮	⋮
13 [R/S]	U=66.00
[R/S]	Z=1.71

User Instructions

SIZE: 004

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Key in the program and set USER mode.		[USER]	
2	Initialize the program.		[XEQ] MANN	MANN-WHITNEY
				N1 ?
3	Input n_1,	n_1	[R/S]	N2 ?
	and n_2.	n_2	[R/S]	R1 ?
4	Perform steps 4-5 for i = 1, 2, ..., n .			
	Input R_i.	R_i	[R/S]	R(i+1) ?
5	If you make a mistake inputting R_h, delete			
	it and go to step 4.	R_h as entered	[C]	R(h) ?
6	U is calculated automatically after the			U=(u)
	n 'th input. Calculate Z.		[R/S]	Z=(z)

Program Listings

Initialize

```
01♦LBL "MAN
N"
02 CF 00
03 FIX 0
04 CLRG
05 CF 29
06 "MANN-WH
ITNEY"
07 AVIEW
08 PSE
```

Input n_1, n_2

```
09 "N1 ?"
10 PROMPT
11 STO 00
12 "N2 ?"
13 PROMPT
14 STO 02
15 GTO A
```

Correction routine

```
16♦LBL C
17 SF 00
18 ST- 03
```

Input R_i

```
19♦LBL A
20 1
21 FS?C 00
22 CHS
23 ST+ 01
24 RCL 00
25 RCL 01
26 X>Y?
27 GTO E
28 "R"
29 ARCL 01
30 "⊢ ?"
31 PROMPT
32 ST+ 03
33 GTO A
```

Calculate U

```
34♦LBL E
35 FIX 2
36 RCL 01
37 2
38 /
39 +
40 RCL 00
41 *
42 RCL 03
43 -
44 "U"
45 XEQ 11
```

Calculate Z

```
46 RCL 00
47 RCL 02
48 *
49 2
50 /
51 -
52 RCL 01
53 RCL 02
54 +
55 RCL 00
56 *
57 RCL 02
58 *
59 12
60 /
61 SQRT
62 /
63 "Z"
```

Display routine

```
64♦LBL 11
65 "⊢="
66 ARCL X
67 AVIEW
68 STOP
69 RTN
70 .END.
```

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	n_1	50	
	n_1+1		
	n_2		
	ΣR_i		
05		55	
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 004 TOT. REG. 22 USER MODE

ENG ___ FIX 2 SCI ___ ON X OFF ___

DEG ___ RAD ___ GRAD ___

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
00		Correction	Normal
29		For proper display format	

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

KENDALL'S COEFFICIENT OF CONCORDANCE

Suppose n individuals are ranked from 1 to n according to some specified characteristic by k observers, the coefficient of concordance W measures the agreement between observers (or concordance between rankings).

$$W = \frac{12 \sum_{i=1}^{n} \left(\sum_{j=1}^{k} R_{ij} \right)^2}{k^2 \, n(n^2 - 1)} - \frac{3(n + 1)}{n - 1}$$

Where R_{ij} is the rank assigned to the ith individual by the jth observer.

W varies from 0 (no community of preference) to 1 (perfect agreement). The null hypothesis that the observers have no community of preference may be tested using special tables, or if $n > 7$, by calculating

$$\chi^2 = k \, (n - 1) \, W$$

which has approximately the chi-aquare distribution with n - 1 degrees of freedom (df).

Operating Limits and Warnings:

For small samples (say, less than or equal to 7) the specially constructed tables should be used. For example: Rank Correlation Methods, M.G. Kendall, Hafner Publishing Co., 1962.

References: 1. Nonparametric Statistical Inference, J. D. Gibbond, McGraw-Hill, 1971.

2. This program is a translation of the HP-65 Stat Pac 1 program.

Example:

Find W, χ^2, and df for the following data:

Table for R_{ij} ($n = 4$, $k = 3$)

i \ j	1	2	3
1	6	7	3
2	1	4	2
3	9	3	5
4	2	6	1

Keystrokes:	Display:
[USER]	(set USER mode)
[XEQ] [ALPHA] SIZE [ALPHA] 007	
[XEQ] [ALPHA] KEN [ALPHA]	KENDALLS COF.
	K?
3 [R/S]	R1,1 ?
6 [R/S]	R1,2 ?
7 [R/S]	R1,3 ?
3 [R/S]	S1=16
[R/S]	R2,1 ?
1 [R/S]	R2,2 ?
⋮	⋮
⋮	R4,3 ?
1 [R/S]	S4=9
[E]	W=10.00
[R/S]	CHI SQ=90.00
[R/S]	dF=3.00

NOTE: Although this example violates the warning ($n < 7$), the amount of data to be entered has been kept small to allow the user to run through the example in short order.

User Instructions

SIZE: 007

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Key in the program and set USER mode.		[USER]	
2	Initialize the program.		[XEQ] KEN	KENDALLS COF.
				K?
3	Input k.	k	[R/S]	R1,1 ?
4	Input R_{ij}. Repeat steps 4-5 for			
	i = 1, 2, ..., k.	R_{ij}	[R/S]	R(i),(j+1)
5	If you make a mistake inputting R_{ih},			
	delete it and go to step 4.	R_{ih}	[C]	R(i),(h)?
6	The sum of the i'th row is automatically			Si=(ΣR_{ij})
	calculated when $R_{i,k}$ is input. Press		[R/S]	R(i+1),1 ?
	[R/S] to continue, or calculate W,		[E]	W=(W)
	χ^2,		[R/S]	CHI SQ=(χ^2)
	and df.		[R/S]	dF=(df)
7	For another set of data, go to step 2.			

Program Listings

Initialize

```
01◆LBL "KEN"
02 CLRG
03 FIX 0
04 CF 29
05 "KENDALLS COF."
06 AVIEW
07 PSE
08 "K?"
09 PROMPT
10 STO 05
11 GTO A
```

Correction routine

```
12◆LBL C
13 ST- 02
14 1
15 ST- 01
```

Accumulate sums

```
16◆LBL A
17 "R"
18 RCL 01
19 1
20 +
21 RCL 04
22 1
23 +
24 ARCL X
25 "⊢,"
26 ARCL Y
27 "⊢ ?"
28 PROMPT
29 ST+ 02
30 1
31 ST+ 01
32 RCL 01
33 RCL 05
34 X>Y?
35 GTO A
36 1
37 ST+ 04
38 RCL 02
39 STO 06
40 X↑2
41 ST+ 03
42 0
43 STO 01
44 STO 02
45 RCL 06
46 "S"
47 "⊢"
48 ARCL 04
49 XEQ 11
50 GTO A
```

Calculate W

```
51◆LBL E
52 FIX 2
53 RCL 03
54 12
55 *
56 RCL 05
57 X↑2
58 /
59 RCL 04
60 /
61 RCL 04
62 X↑2
63 1
64 -
65 /
66 RCL 04
67 1
68 ST- 04
69 +
70 3
71 *
72 RCL 04
73 /
74 -
75 "W"
76 XEQ 11
```

Calculate χ^2 and df

```
77 RCL 05
78 *
79 RCL 04
80 *
81 "CHI SQ"
82 XEQ 11
83 RCL 04
84 "dF"
```

Display routine

```
85◆LBL 11
86 "⊢="
87 ARCL X
88 AVIEW
89 STOP
90 RTN
91 .END.
```

00

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00		50	
	j ... k		
	ΣR_{ij}		
	$\Sigma (R_{ij})^2$		
	i ... n		
05	K	55	
	ΣR_{ij}		
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 007 TOT. REG. 30 USER MODE
ENG ____ FIX 2 SCI ____ ON X OFF ____
DEG ____ RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
29		For proper display format	

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

HEWLETT-PACKARD

HP-41C

USERS' LIBRARY SOLUTIONS

Bar Codes

Test Statistics

TEST STATISTICS

ONE SAMPLE TEST STATISTICS FOR THE MEAN 1
TEST STATISTICS FOR THE CORRELATION COEFFICIENT 2
DIFFERENCES AMONG PROPORTIONS 3
BEHRENS-FISHER STATISTIC 4
KRUSKAL-WALLIS STATISTIC 5
MEAN-SQUARE SUCCESSIVE DIFFERENCE 6
THE RUN TEST FOR RANDOMNESS 7
INTRACLASS CORRELATION COEFFICIENT 8
FISHER'S EXACT TEST FOR A 2 X 2 CONTINGENCY TABLE 9
BARTLETT'S CHI-SQUARE STATISTIC 10
MANN-WHITNEY STATISTIC 11
KENDALL'S COEFFICIENT OF CONCORDANCE 12

NOTICE

ONE SAMPLE TEST
STATISTICS FOR THE MEAN
PROGRAM REGISTERS NEEDED: 16

ROW 1 (1 - 4)

ROW 2 (4 - 5)

ROW 3 (5 - 9)

ROW 4 (9 - 12)

ROW 5 (12 - 23)

ROW 6 (24 - 31)

ROW 7 (31 - 36)

ROW 8 (37 - 42)

ROW 9 (43 - 45)

TEST STATISTICS FOR
THE CORRELATION COEFFICIENT
PROGRAM REGISTERS NEEDED: 19

ROW 1 (1 - 3)

ROW 2 (3 - 3)

ROW 3 (3 - 11)

ROW 4 (12 - 18)

ROW 5 (19 - 29)

ROW 6 (30 - 31)

ROW 7 (31 - 40)

ROW 8 (41 - 53)

ROW 9 (54 - 62)

ROW 10 (63 - 74)

ROW 11 (75 - 75)

DIFFERENCES AMONG PROPORTIONS

PROGRAM REGISTERS NEEDED: 26

ROW 1 (1 - 4)

ROW 2 (5 - 6)

ROW 3 (6 - 11)

ROW 4 (12 - 18)

ROW 5 (19 - 26)

ROW 6 (27 - 37)

ROW 7 (38 - 48)

ROW 8 (48 - 55)

ROW 9 (55 - 65)

ROW 10 (66 - 71)

ROW 11 (71 - 79)

ROW 12 (80 - 85)

ROW 13 (85 - 93)

RCW 14 (93 - 98)

BEHRENS–FISHER STATISTIC

PROGRAM REGISTERS NEEDED: 19

ROW 1 (1 – 5)

ROW 2 (5 – 6)

ROW 3 (6 – 15)

ROW 4 (16 – 19)

ROW 5 (20 – 30)

ROW 6 (30 – 35)

ROW 7 (35 – 40)

ROW 8 (41 – 51)

ROW 9 (51 – 58)

ROW 10 (59 – 65)

KRUSKAL–WALLIS STATISTIC

PROGRAM REGISTERS NEEDED: 20

ROW 1 (1 – 5)

ROW 2 (5 – 5)

ROW 3 (6 – 12)

ROW 4 (13 – 21)

ROW 5 (22 – 28)

ROW 6 (29 – 35)

ROW 7 (36 – 45)

ROW 8 (45 – 55)

ROW 9 (56 – 65)

ROW 10 (66 – 71)

ROW 11 (72 – 76)

MEAN–SQUARE
SUCCESSIVE DIFFERENCE
PROGRAM REGISTERS NEEDED: 15

ROW 1 (1 – 4)

ROW 2 (4 – 6)

ROW 3 (6 – 15)

ROW 4 (16 – 26)

ROW 5 (27 – 37)

ROW 6 (37 – 46)

ROW 7 (47 – 56)

ROW 8 (57 – 61)

THE RUN TEST
FOR RANDOMNESS
PROGRAM REGISTERS NEEDED: 20

ROW 1 (1 – 2)

ROW 2 (3 – 6)

ROW 3 (6 – 7)

ROW 4 (8 – 9)

ROW 5 (9 – 12)

ROW 6 (12 – 20)

ROW 7 (21 – 29)

ROW 8 (30 – 42)

ROW 9 (43 – 48)

ROW 10 (49 – 57)

ROW 11 (58 – 59)

INTRACLASS
CORRELATION COEFFICIENT
PROGRAM REGISTERS NEEDED: 28

ROW 1 (1 - 5)

ROW 2 (5 - 6)

ROW 3 (6 - 12)

ROW 4 (13 - 21)

ROW 5 (22 - 28)

ROW 6 (28 - 36)

ROW 7 (36 - 44)

ROW 8 (45 - 54)

ROW 9 (55 - 60)

ROW 10 (60 - 72)

ROW 11 (73 - 85)

ROW 12 (86 - 94)

ROW 13 (95 - 104)

ROW 14 (105 - 110)

ROW 15 (110 - 117)

FISHER'S EXACT TEST FOR A
2 X 2 CONTINGENCY TABLE
PROGRAM REGISTERS NEEDED: 25

ROW 1 (1 - 4)

ROW 2 (5 - 5)

ROW 3 (6 - 14)

ROW 4 (15 - 23)

ROW 5 (24 - 36)

ROW 6 (37 - 49)

ROW 7 (50 - 62)

ROW 8 (62 - 69)

ROW 9 (69 - 76)

ROW 10 (76 - 83)

ROW 11 (84 - 90)

ROW 12 (90 - 98)

ROW 13 (98 - 105)

RCW 14 (105 - 105)

BARTLETT'S CHI-SQUARE
STATISTIC
PROGRAM REGISTERS NEEDED: 24

ROW 1 (1 - 5)

ROW 2 (5 - 8)

ROW 3 (9 - 16)

ROW 4 (16 - 24)

ROW 5 (25 - 33)

ROW 6 (34 - 41)

ROW 7 (41 - 48)

ROW 8 (48 - 56)

ROW 9 (57 - 64)

ROW 10 (64 - 76)

ROW 11 (77 - 87)

ROW 12 (87 - 92)

ROW 13 (92 - 97)

MANN-WHITNEY STATISTIC

PROGRAM REGISTERS NEEDED: 19

ROW 1 (1 - 4)

ROW 2 (5 - 6)

ROW 3 (6 - 12)

ROW 4 (12 - 18)

ROW 5 (18 - 27)

ROW 6 (27 - 32)

ROW 7 (33 - 41)

ROW 8 (42 - 51)

ROW 9 (52 - 63)

ROW 10 (63 - 70)

KENDALL'S COEFFICIENT
OF CONCORDANCE
PROGRAM REGISTERS NEEDED: 24

ROW 1 (1 - 5)

ROW 2 (5 - 5)

ROW 3 (6 - 13)

ROW 4 (13 - 22)

ROW 5 (23 - 28)

ROW 6 (29 - 37)

ROW 7 (37 - 47)

ROW 8 (47 - 52)

ROW 9 (53 - 64)

ROW 10 (65 - 75)

ROW 11 (76 - 81)

ROW 12 (81 - 87)

ROW 13 (87 - 91)

NOTES

HEWLETT PACKARD

Rev. A

Hewlett-Packard Software

In terms of power and flexibility, the problem-solving potential of the HP-41C programmable calculator is nearly limitless. And in order to see the practical side of this potential, HP has different types of software to help save you time and programming effort. Every one of our software solutions has been carefully selected to effectively increase your problem-solving potential. Chances are, we already have the solutions you're looking for.

Application Pacs

To increase the versatility of your HP-41C, HP has an extensive library of "Application Pacs". These programs transform your HP-41C into a specialized calculator in seconds. Included in these pacs are detailed manuals with examples, minature plug-in Application Modules, and keyboard overlays. Every Application Pac has been designed to extend the capabilities of the HP-41C.

You can choose from:

Aviation
Clinical Lab
Circuit Analysis
Financial Decisions
Mathematics
Structural Analysis
Surveying
Securities
Statistics
Stress Analysis
Games
Home Management
Machine Design
Navigation
Real Estate
Thermal and Transport Science

Users' Library

The Users' Library provides the best programs from contributors and makes them available to you. By subscribing to the HP-41C Users' Library you'll have at your fingertips literally hundreds of different programs from many different application areas.

* Users' Library Solutions Books

Hewlett-Packard offers a wide selection of Solutions Books complete with user instructions, examples, and listings. These solution books will complement our other software offerings and provide you with a valuable tool for program solutions.

You can choose from:

Business Stat/Marketing/Sales
Home Construction Estimating
Lending, Saving and Leasing
Real Estate
Small Business
Geometry
High-Level Math
Test Statistics
Antennas
Chemical Engineering
Control Systems
Electrical Engineering
Fluid Dynamics and Hydraulics
Civil Engineering
Heating, Ventilating & Air Conditioning
Mechanical Engineering
Solar Engineering
Calendars
Cardiac/Pulmonary
Chemistry
Games
Optometry I (General)
Optometry II (Contact Lens)
Physics
Surveying

* Some books require additional memory modules to accomodate all programs.

TEST STATISTICS

ONE SAMPLE TEST STATISTICS FOR THE MEAN
TEST STATISTICS FOR THE CORRELATION COEFFICIENT
DIFFERENCES AMONG PROPORTIONS
BEHRENS-FISHER STATISTIC
KRUSKAL-WALLIS STATISTIC
MEAN-SQUARE SUCCESSIVE DIFFERENCE
THE RUN TEST FOR RANDOMNESS
INTRACLASS CORRELATION COEFFICIENT
FISHER'S EXACT TEST FOR A 2×2 CONTINGENCY TABLE
BARTLETT'S CHI-SQUARE STATISTIC
MANN-WHITNEY STATISTIC
KENDALL'S COEFFICIENT OF CONCORDANCE

HEWLETT PACKARD

00041-90082 Printed in U.S.A.